The Proof of the Inverse Shell Theorem

Tigran Aivazian, 6th April 2018.

Statement of the Problem

The direct Shell Theorem was proved by Isaac Newton in his Principia (1687) and states that:

- The net gravitational force at any point inside the hollow sphere is zero.
- The gravitational force at any point outside the hollow sphere is the same as that produced by a point at the centre with the mass equal to the total mass of the sphere.

We shall be more concerned with proving the *inverse* statement, namely these:

- If the net gravitational force at any point inside the hollow sphere is zero, then gravitation obeys the inverse square $1/r^2$ law.
- If the gravitational force at any point outside the hollow sphere is the same as that produced by a point at the centre with the mass equal to the total mass of the sphere, then gravitation obeys the inverse square $1/r^2$ law.

The same statements hold in the case of electrostatic Coulomb field with the obvious difference that the masses are replaced with the electric charges.

Proof : 1. Point Inside the Shell

We begin by calculating the force $\delta\boldsymbol{F}$ exerted by the infinitesemal charged element δq of the sphere on a point at a distance r_0 from the centre of the sphere, as shown on the diagram below:



For convenience, we have oriented the z-axis of our coordinate system along the direction from the centre of the sphere to the point with the test charge $+e$. Then, we can obtain the vector force acting at this charge by summing the contributions from all the charges distributed along the surface of the sphere, as follows:

$$\boldsymbol{R} = \boldsymbol{r}_0 - \boldsymbol{r}$$
$$R^2 = {r_0}^2 + r^2 - 2\, r_0\, r \cos\theta$$
$$\delta \boldsymbol{F} = e\, \delta \mathrm{q}\, f(R)\, \frac{\boldsymbol{R}}{R}$$
$$\delta \mathrm{q} = \frac{Q}{4\pi r^2}\, dS$$
$$dS = r^2 \sin\theta\, d\theta\, d\varphi$$
$$\boldsymbol{F} = \frac{e\,Q}{4\pi} \int_0^{2\pi} d\varphi \int_0^{\pi} \frac{f(R)}{R}\, \boldsymbol{R} \sin\theta\, d\theta$$

First let us prove that $F_x = F_y = 0$:

$$(\boldsymbol{R},\ \boldsymbol{e}_x) = (\boldsymbol{r}_0 - \boldsymbol{r},\ \boldsymbol{e}_x) = -(r,\ \boldsymbol{e}_x) = -r \sin\theta \cos\varphi$$
$$F_x \propto \int_0^{2\pi} \cos\varphi\, d\varphi = 0$$

and likewise for $F_y = 0$. Further, we obtain the only non-vanishing component of the force, viz. F_z:

$(\boldsymbol{R},\ \boldsymbol{e}_z) = (\boldsymbol{r}_0 - \boldsymbol{r},\ \boldsymbol{e}_z) = r_0 - r\cos\theta,\ \ x = \cos\theta$

$F_z = \frac{e\,Q\,r_0}{2} \int\limits_0^{\pi} \frac{f(R)}{R} \left(1 - \frac{r\cos\theta}{r_0}\right) \sin\theta\, d\theta = \frac{e\,Q\,r_0}{2} \int\limits_{-1}^{+1} \left(1 - \frac{r\,x}{r_0}\right) \frac{f(R)}{R}\,\mathrm{dx}$

Let us convert integration over x to the integration over R:

$2\,R\,dR = -2\,r_0\,r\,dx$

$x = \frac{r_0{}^2 + r^2 - R^2}{2\,r_0\,r}$

$F_z = \frac{e\,Q}{4\,r\,r_0{}^2} \int\limits_{r-r_0}^{r+r_0} \left(r_0{}^2 - r^2 + R^2\right) f(R)\,dR$

Note that the above formula for F_z is only valid inside the sphere $r_0 < r$. For the point outside the sphere the formula would be:

$F_z = \frac{e\,Q}{4\,r\,r_0{}^2} \int\limits_{r_0-r}^{r_0+r} \left(r_0{}^2 - r^2 + R^2\right) f(R)\,dR$

Now we need to prove that if the above expression is equal to 0 for all the values of r_0 and r , then the function $f(R)$ necessarily has the form of inverse square law $1/r^2$.

First we calculate the Taylor series of the integrand near the point $R = r$ up to the sixth degree:

```
In[1]:= series = Series[(r0^2 - r^2 + R^2) * f[R], {R, r, 6}]
```

Out[1]=

$$r0^2\, f[r] + \left(2\, r\, f[r] + r0^2\, f'[r]\right)(R - r) + \left(f[r] + 2\, r\, f'[r] + \frac{1}{2}\, r0^2\, f''[r]\right)(R - r)^2 + \left(f'[r] + r\, f''[r] + \frac{1}{6}\, r0^2\, f^{(3)}[r]\right)(R - r)^3 + \frac{1}{24}\left(12\, f''[r] + 8\, r\, f^{(3)}[r] + r0^2\, f^{(4)}[r]\right)(R - r)^4 + \frac{1}{120}\left(20\, f^{(3)}[r] + 10\, r\, f^{(4)}[r] + r0^2\, f^{(5)}[r]\right)(R - r)^5 + \frac{1}{720}\left(30\, f^{(4)}[r] + 12\, r\, f^{(5)}[r] + r0^2\, f^{(6)}[r]\right)(R - r)^6 + O[R - r]^7$$

Then we calculate the integral with the assumption that the point lies inside the sphere:

```
In[2]:= int = Integrate[series, {R, r - r0, r + r0},
            Assumptions → {r > r0 > 0}] // Normal
```

Out[2]=
$$\frac{1}{2520} r0^3 \left(6720\, f[r] + 3360\, r\, f'[r] + r0^2 \left(1344\, f''[r] + 336\, r\, f^{(3)}[r] + r0^2 \left(72\, f^{(4)}[r] + 12\, r\, f^{(5)}[r] + r0^2\, f^{(6)}[r]\right)\right)\right)$$

We need to collect the coefficients corresponding to the powers of r_0:

```
In[3]:= coef = CoefficientList[1 / (r0 ^ 2) * int, r0] // Factor
```

Out[3]=
$$\left\{0,\ \frac{4}{3}\left(2\, f[r] + r\, f'[r]\right),\ 0,\ \frac{2}{15}\left(4\, f''[r] + r\, f^{(3)}[r]\right),\ 0,\ \frac{1}{210}\left(6\, f^{(4)}[r] + r\, f^{(5)}[r]\right),\ 0,\ \frac{f^{(6)}[r]}{2520}\right\}$$

But we don’t need zeros, so we tidy up the above list (dropping the last element, because it belongs to the elements of the series of higher degrees than we allowed (six):

```
In[4]:= coef = DeleteCases[coef // Factor // Most, 0]
```

Out[4]=
$$\left\{\frac{4}{3}\left(2\, f[r] + r\, f'[r]\right),\ \frac{2}{15}\left(4\, f''[r] + r\, f^{(3)}[r]\right),\ \frac{1}{210}\left(6\, f^{(4)}[r] + r\, f^{(5)}[r]\right)\right\}$$

Now we need to demand that all these coefficients are zero identically and solve the corresponding differential equations, storing the solutions in the variable sols:

```
In[5]:= sols = Table[
          f[r] /. Flatten@DSolve[Thread[# == 0 &@coef][[k]], f[r], r,
            GeneratedParameters → (Subscript[c, #, k] &)], {k, 3}]
```

Out[5]=
$$\left\{\frac{c_{1,1}}{r^2},\ \frac{c_{1,2}}{6\, r^2} + c_{2,2} + r\, c_{3,2},\ \frac{c_{1,3}}{120\, r^2} + c_{2,3} + r\, c_{3,3} + r^2\, c_{4,3} + r^3\, c_{5,3}\right\}$$

We can use Mathematica’s function SolveAlways[] to find the set of integration constants which make the above three solutions coincide:

```
In[6]:= solset = sols /. SolveAlways[Equal @@@ Subsets[sols, {2}], r]
```

Out[6]=
$$\left\{\left\{\frac{c_{1,3}}{120\, r^2},\ \frac{c_{1,3}}{120\, r^2},\ \frac{c_{1,3}}{120\, r^2}\right\}\right\}$$

We see already that we have proved what we wanted, but we can further tidy up the result cosmetically, by getting rid of duplicates and renaming the constant of integration:

```
In[7]:= Union @@@ solset /. Subscript[c, 1, 3] / 120 → c
```

Out[7]= $\left\{\frac{c}{r^2}\right\}$

Thus we have proved the first part of the Inverse Shell Theorem, namely we have established that *only* the inverse square law has the property of vanishing gravitational/electrostatic field inside the hollow sphere.

Let us verify this proof by substituting this function in the main integral – we expect to obtain zero as a result:

```
In[8]:= Integrate[(r0^2 - r^2 + R^2) * c / R^2,
          {R, r - r0, r + r0}, Assumptions → {r > r0 > 0}]
```

Out[8]= 0

Proof : 2. Point Outside the Shell

We already have the expression for the force acting at a point outside the sphere:

$$F_z = \frac{e\,Q}{4\,r\,{r_0}^2} \int_{r_0-r}^{r_0+r} \left({r_0}^2 - r^2 + R^2\right) f(R)\, dR$$

we demand that this is equal to the usual Coulomb force originating from the centre with the total charge Q:

$$F_z = \frac{e\,Q}{{r_0}^2}$$

leading to the following non-homogeneous integral equation for $f(R)$:

$$\int_{r_0-r}^{r_0+r} \left({r_0}^2 - r^2 + R^2\right) f(R)\, dR = 4\,r$$

Now we begin by calculating the series of the integrand in the neighbourhood of the point $R = r_0$:

```
In[9]:= series2 = Series[(r0^2 - r^2 + R^2) * f[R], {R, r0, 6}]
```

Out[9]=

$$\left(-r^2 f[r0] + 2\, r0^2 f[r0]\right) + \left(2\, r0\, f[r0] - r^2 f'[r0] + 2\, r0^2 f'[r0]\right) (R - r0) + \left(f[r0] + 2\, r0\, f'[r0] - \frac{1}{2} r^2 f''[r0] + r0^2 f''[r0]\right) (R - r0)^2 + \left(f'[r0] + r0\, f''[r0] - \frac{1}{6} r^2 f^{(3)}[r0] + \frac{1}{3} r0^2 f^{(3)}[r0]\right) (R - r0)^3 + \frac{1}{24}\left(12 f''[r0] + 8\, r0\, f^{(3)}[r0] - r^2 f^{(4)}[r0] + 2\, r0^2 f^{(4)}[r0]\right) (R - r0)^4 + \frac{1}{120}\left(20 f^{(3)}[r0] + 10\, r0\, f^{(4)}[r0] - r^2 f^{(5)}[r0] + 2\, r0^2 f^{(5)}[r0]\right) (R - r0)^5 + \frac{1}{720}\left(30 f^{(4)}[r0] + 12\, r0\, f^{(5)}[r0] - r^2 f^{(6)}[r0] + 2\, r0^2 f^{(6)}[r0]\right) (R - r0)^6 + O[R - r0]^7$$

As before, we proceed to calculate the integral:

```
In[10]:= int2 = Integrate[series2, {R, r0 - r, r0 + r},
            Assumptions → {r0 > r > 0}] // Normal
```

Out[10]=

$$-\frac{4}{3}\left(r^3 - 3\, r\, r0^2\right) f[r0] + \frac{1}{2520} r^3 \left(3360\, r0\, f'[r0] - 336\left(r^2 - 5\, r0^2\right) f''[r0] + r^2\left(336\, r0\, f^{(3)}[r0] - 12\left(r^2 - 7\, r0^2\right) f^{(4)}[r0] + r^2\left(12\, r0\, f^{(5)}[r0] - \left(r^2 - 2\, r0^2\right) f^{(6)}[r0]\right)\right)\right)$$

And then calculate and tidy up the coefficients at the power of *r*:

```
In[11]:= coef2 = CoefficientList[int2 - 4 * r, r]
```

Out[11]=

$$\left\{0,\ -4 + 4\, r0^2 f[r0],\ 0,\ -\frac{4 f[r0]}{3} + \frac{4}{3} r0\, f'[r0] + \frac{2}{3} r0^2 f''[r0],\ 0,\ -\frac{2}{15} f''[r0] + \frac{2}{15} r0\, f^{(3)}[r0] + \frac{1}{30} r0^2 f^{(4)}[r0],\ 0,\ -\frac{1}{210} f^{(4)}[r0] + \frac{1}{210} r0\, f^{(5)}[r0] + \frac{r0^2 f^{(6)}[r0]}{1260},\ 0,\ -\frac{f^{(6)}[r0]}{2520}\right\}$$

In[12]:=
```
coef2 = DeleteCases[coef2 // Factor // Most, 0]
```

Out[12]=
$$\left\{4\left(-1+r0^2\,f[r0]\right),\ -\frac{2}{3}\left(2\,f[r0]-2\,r0\,f'[r0]-r0^2\,f''[r0]\right),\ \frac{1}{30}\left(-4\,f''[r0]+4\,r0\,f^{(3)}[r0]+r0^2\,f^{(4)}[r0]\right),\ \frac{-6\,f^{(4)}[r0]+6\,r0\,f^{(5)}[r0]+r0^2\,f^{(6)}[r0]}{1260}\right\}$$

In[13]:=
```
sols2 = Table[f[r0] /.
    Flatten@DSolve[Thread[# == 0 &@coef2][[k]], f[r0], r0,
      GeneratedParameters → (Subscript[c, #, k] &)], {k, 3}]
```

Out[13]=
$$\left\{\frac{1}{r0^2},\ r0\,c_{1,2}+\frac{c_{2,2}}{r0^2},\ \frac{1}{6}\left(r0^3\,c_{1,3}+\frac{c_{2,3}}{r0^2}\right)+c_{3,3}+r0\,c_{4,3}\right\}$$

In[14]:=
```
solset =
  sols2 /. SolveAlways[Equal @@@ Subsets[sols2, {2}], r0]
```

Out[14]=
$$\left\{\left\{\frac{1}{r0^2},\ \frac{1}{r0^2},\ \frac{1}{r0^2}\right\}\right\}$$

Finally, we get rid of duplicates by calling Union[]:

In[15]:=
```
Union[solset[[1]]]
```

Out[15]=
$$\left\{\frac{1}{r0^2}\right\}$$

Now we have to verify our solution by direct substitution into the original integral expression:

In[16]:=
```
Integrate[(r0^2 - r^2 + R^2) * 1 / R^2,
   {R, r0 - r, r0 + r}, Assumptions → {r0 > r > 0}] - 4 * r
```

Out[16]= 0

Thus we have proved both statements of the Inverse Shell Theorem.

References

I asked the question about solving the integral equation in the first section of this paper on the Mathematica Stackexchange Forum and a very knowledgeable user called Artes replied describing the method used here, which I have also applied to the second part of the problem:

https://mathematica.stackexchange.com/questions/170445/how-do-i-solve-a-kind-of-integral-equation-with-mathematica